CEWES MSRC/PET TR/99-30

Fortran 90 Namelist I/O versus Cray Namelist I/O

by

Dan Nagle
Joy Brogdon

05h00699

Work funded by the DoD High Performance Computing Modernization Program CEWES Major Shared Resource Center through

Programming Environment and Training (PET)

Supported by Contract Number: DAHC94-96-C0002
Nichols Research Corporation

Fortran 90 Namelist I/O versus Cray Namelist I/O

Dan Nagle and Joy Brogdon

Migrating serial codes to parallel systems can be an intimidating endeavor. Compounding the programmer's apprehension may be the prospect of using Fortran 90 instead of Fortran 77.

Fortran 90 standardized namelist input/output (I/O). Cray Fortran supported namelist as an extension prior to the Fortran 90 standard. There are some differences between the older Cray namelist and the Fortran 90 namelist. This report will discuss the most important differences. For a more complete description, the interested reader is referred to the Cray publication *Fortran Language Reference Manual, Volume 3* (publication number SR-3905, Edition 3.0, pages 93 et seq.).

As a vendor extension, namelist was defined in an ad hoc way. The Fortran Standards Committee defined namelist to be consistent with the free-format specification. Therefore, there are some differences between the two which require the programmer's attention prior to using the older Cray namelist files with standard-conforming programs. Note that Cray's Fortran 90 compiler supports the Fortran 90 standard, but also accepts the older extensions, at least under certain circumstances. In general, the best course of action is to change the input files to adhere to the standard. Any output files written by a Fortran 90 compiler will, for the most part, be written in a Fortran 90 standard-conforming way by default.

The Form of the Namelist Record

Both standard namelist and the older extension allowed namelist records to be continued on several lines. The standard form of a namelist input record is:

```
&group_name [ name = value [,]] ... /
```

The `&group_name` must be the first non-blank characters in the record being read, and must be the first non-blank characters on the line. The "/" character terminates the namelist input record; it is also the same character that terminates a free-format input record.

The earlier Cray extension had, by default, the following form:

```
&group_name [ name = value [,]] ... &END
```

The `&group_name` was required to occur after column one, typically was the first non-blank characters in the record being read, and the "&" had to be the first non-blank character after column one on the line. If the character "E" appeared in column one, it was an echo flag that caused the input record to be echoed to standard output. Note that the echo character could be changed by the program. The input record was ended by the "&END" characters, not by the "/" character, as in the standard. Therefore, a standard-conforming program reading a Cray namelist record will probably fail

when the program encounters the end of file without finding the terminating “/” character.

The older Cray extension also accepted the "$" character in place of the "&" character (see example program below). Thus, the following statement was an acceptable format for a Cray namelist:

```
$group_name [ name = value [,]] ... $END
```

This format is not acceptable to a standard-conforming compiler.

```
      PROGRAM EXAMPLE   ! typical NAMELIST usage in Cray Fortran 77
      LOGICAL DONE
      REAL CTEMP
      DATA FTEMP,DONE /75.0,.FALSE./
      NAMELIST /INPUT/ FTEMP,DONE
      NAMELIST /OUTPUT/ CTEMP

      OPEN (UNIT=5,FILE='NML.77',FORM='FORMATTED',STATUS='OLD')

10    READ(5,INPUT)
      IF (DONE) STOP
      CTEMP = ((FTEMP - 32.0) / 1.8)
      PRINT *,CTEMP
      GOTO 10
      END
```

The example above takes the following input file, NML.77:

```
&INPUT &END
&INPUT FTEMP=212.0 &END
&INPUT FTEMP=32.0 &END
&INPUT DONE=.TRUE. &END
```

The programmer should change such input records to be standard-conforming. The following example namelist adheres to the Fortran 90 standard:

```
&INPUT /
&INPUT FTEMP=212.0 /
&INPUT FTEMP=32.0 /
&INPUT DONE=.TRUE. /
```

Comments in Namelist Input Records

The Fortran 90 standard (as opposed to the Fortran 95 standard) allows comments in namelists. Similar to source code, the comment begins at the "!" character and extends to the end of the line. In the earlier Cray namelist, comments begin at either the ";" character or the ":" character and extended to end of the line. The programmer could also change the comment character in Cray namelists. In standard Fortran, the comment character cannot be changed.

Cray Namelist Auxiliary Routines

The earlier Cray namelist supported a number of auxiliary routines that allowed the programmer to modify namelist format and behavior. The routines that affect namelist input are summarized below:

`RNLSKIP()` Controlled the action taken if the next namelist group name encountered in the file was not the next one to be read.
`RNLTYPE()` Controlled the action taken when type mismatches occurred.
`RNLECHO()` Set the unit to receive echoed input records.
`RNLFLAG()` Set characters to be used to indicate echoing of input records, if present in column one.
`RNLDELM()` Set characters to precede the namelist group name and the "END" string.
`RNLSEP()` Set characters to be used as separators between namelist name-value pairs.
`RNLREP()` Set characters to be used to separate the variable name from the value.
`RNLCOMM()` Set characters to be used to start a comment in a namelist input record.

In general, the best course of action when any of these is found in a program is to eliminate the reference, and modify the namelist input record to be standard-conforming. The Cray namelist auxiliary routines that affect namelist output are listed below:

`WNLLONG()` Set the maximum output line length.
`WNLDELM()` Set the character preceding the namelist group name and the "END" string.
`WNLSEP()` Set characters to be used as separators between namelist name-value pairs.
`WNLFLAG()` Set characters to be placed in column one of output records.
`WNLLINE()` Controlled whether each name-value pair occurred on a separate line.

Since these procedures are all non-standard, it is best to ignore them and use standard-conforming output from namelist write statements. Note in particular that the `RNMDELM()` and `WNMDELM()` routines may be used to convert older Cray namelist input records into standard-conforming records. The routines add characters that are used to precede both the namelist group name and the terminating "END" string. The addition of the "/" to the list of characters which precede both the group name and the "END" string, the program could read a standard-conforming namelist record with the older Cray namelist. The price, however, is one of portability to other systems.

Using Cray Namelist Format on the IBM SP

It is possible to use old-style Cray namelist format on the IBM SP by inserting the following call in the source code:

```
call setrteopts( "namelist=old" )
```

See the following example.

```
PROGRAM EXAMPLE         ! Old format NAMELIST usage on IBM
LOGICAL DONE
REAL CTEMP
DATA FTEMP,DONE /75.0,.FALSE./
```

```
      NAMELIST /INPUT/ FTEMP,DONE
      NAMELIST /OUTPUT/ CTEMP

      OPEN (UNIT=5,FILE='NML.77',FORM='FORMATTED',STATUS='OLD')

      call setrteopts( "namelist=old" )

10    READ(5,INPUT)
      IF (DONE) STOP
      CTEMP = ((FTEMP - 32.0) / 1.8)
      PRINT *,CTEMP
      GOTO 10
      END
```

However, it is probably as easy to change the "&END" to "/" in the namelist file. This avoids changing the program source code, and more importantly, maintains portability.